CHARTING INDICATORS

Going Head To Head

Which Trend Indicator Wins?

Trends are obvious in retrospect, but identifying them in real time is a different story. Here, we look at four existing trend detection indicators to see how well they do their job. Which one comes out ahead?

arkets can either trend or move in a trading range (consolidate). Each phase requires the use of different types of indicators. In trending markets, you need to use *trend-following* indicators, such as moving averages, MACD, and so on, whereas for markets that are in a trading range, you need to use *oscillators*, such as the RSI or stochastics, which use overbought and oversold levels.

AVOID GETTING CAUGHT

One problem with all trending systems is that they only work when the market is trending, which is only about 30% of the time. If you fail to recognize that market conditions are not appropriate for trend system trading, you will be whipsawed out of trades with small to moderate losses, which over time will result in significant losses. On the other hand, in trending markets, prices can remain overbought or oversold for extended periods, and an oscillator-based system will produce premature signals. This is why it's critical to identify the market phase when choosing which system to use.

The idea of a system that stays in line with the market by switching from trending indicators to trading range indicators may sound like the closest thing to the holy grail, but implementing such a system in practice may be a challenge. That's because trends are difficult to identify in real time.

The conundrum is that you can't identify a trend until it is well under way, since that is precisely how a trend is defined. The sooner you recognize the trend, the more subject you are to whipsaw reversals. The longer you wait, the more profits you leave on the table on both ends of the trade.

Ideally, such an indicator should not be too late or too sensitive. Realistically, this can only be achieved with some serious concessions, and even though several trend-following indicators are available, the search for such an indicator may seem like a never-ending quest. The most popular and oldest trend-following indicator is the average directional movement index (ADX), but other technicians have developed additional indicators using completely different methods. To determine their efficiency in detecting trends in a timely manner, I tested four popular trend detection indicators. I'll go over each one.

Wilder's average directional index (ADX)

This is by far the most popular trend detection indicator. J. Welles Wilder Jr. presented the ADX in his 1978 book *New Concepts In Technical Trading Systems*. The *plus directional movement* (+DM) and *minus directional movement* (-DM) form the backbone of the ADX.

Directional movement is defined as the difference between the extreme of the current period that falls outside the range of the previous period; price action above yesterday's high is positive directional movement (+DM), while anything below yesterday's low is negative directional movement (-DM).

The directional indicators PDI(+DI) and MDI(-DI) are then calculated by taking the +DM and -DM average over a number of periods (Wilder originally used 14 days), smoothing them using Wilder's special technique, and normalizing the results by dividing by the average true range (ATR).

The directional movement (DX) is then calculated by taking the difference between PDI and MDI and normalizing it in order to express it between zero and 100 according to the following formula:

PATRICK KELLEY

by Markos Katsanos

$$DX = 100 \times \frac{ABS(PDI(14) - MDI(14))}{PDI(14) + MDI(14)}$$

Finally, the ADX is calculated as follows: The first ADX value is simply a 14-day average of DX. Subsequent ADX values are smoothed by multiplying the previous 14-day ADX value by 13, adding the most recent DX value, and dividing this total by 14. ADX readings below 20 indicate trend weakness while readings above 30 indicate trend strength.

Vertical horizontal filter (VHF)

The vertical horizontal filter (VHF) was introduced by Adam White in 1991. The VHF is calculated by dividing the range over a given period (the default is 28 days) by the sum of the absolute value of the difference between consecutive daily closes for the last 28 days. Here's the formula:

$$VHF = \frac{Highest(C,28) - Lowest(C,28)}{\sum ABS(C_i - C_{i-1})}$$

The numerator is the difference between the highest close and the lowest close and the denominator is the absolute sum of the price changes during the lookback period.

The VHF is included in almost all technical and charting software. You should keep in mind, however, that some software, such as TradeStation, calculates the range by taking the difference between the highest high and the lowest low, resulting in higher VHF values. A VHF reading above 0.35 to 0.38 indicates a trending market. This threshold, however, is not standard across all markets and may fluctuate according to the volatility of the security.

Efficiency ratio (ER)

The efficiency ratio (ER) was described by Perry Kaufman in his 1995 book *Smarter Trading*. It is calculated by dividing the net change in price over a given period by the absolute value of the sum of all bar-to-bar price changes over the same period. Here's the formula:

$$ER = \frac{ABS(C - C[Period])}{\sum ABS(C_i - C_{i-1})}$$

The resulting ratio ranges between zero and 1, with higher values representing a more efficient, or trending, market.

As you can see, the formula looks similar to the VHF. The difference is in the numerator, which is the current price change instead of the range during the lookback period. In the comparison test in Figure 1 you will see that this resulted in significant improvement in performance.

R-squared (R^2 or R2)

The *R-squared* or R^2 identifies how closely price is associated with its linear regression line, that is, how close price movement is to a straight line. It is defined as the square of the correlation between price and the linear regression line. So higher R-squared values indicate less deviation between price and the best-fit line. You should therefore use R-squared in tandem with slope. R-squared ranges between zero and 1. High R-squared values accompanied by a steep slope value indicate strong trends. Although there are no standard default parameters for the R-squared period and trend threshold, you should keep in mind that the threshold level is inversely proportional to the period. Therefore, the shorter the period, the higher the threshold should be. In optimizing the parameters in the comparison test, I

IN-SAMPLE PERFORMANCE (2006–2011)						
System	MA	Buy & Hold	ADX	R2	VHF	ER
Net profit	$58,956	$-15,979	$64,789	$65,763	$62,176	$66,900
Total no. of trades	81	1	8	14	10	10
Annual rate of return	17.38%	-7.62%	18.65%	18.86%	18.09%	19.10%
Percent profitable	35.80%	0.00%	75.00%	64.30%	60.00%	63.60%
Profit factor	1.88		17.30	9.78	14.30	21.00
Max losing trade	$-5,247	$-15,979	$-2,864	$-2,196	$-2,864	$-1,345
Max intraday drawdown	$-39,776	$-72,612	$-11,560	$-13,915	$-11,140	$-11,259
TEST PARAMETERS						
MA period	50		50	50	50	50
TDI period			14	18	36	24
TDI smoothing				3		3
TDI trend threshold			30	0.42	0.38	0.36
TDI MAX			42	0.85	0.45	0.42
LRS crit				10		
TDI crit			22		0.24	0.26
LAG			12	10	10	3
MULT			1.8		1.5	2.5

FIGURE 1: IN-SAMPLE (IS) PERFORMANCE OF THE 50-DAY SMA, BUY & HOLD, AND SYSTEMS FILTERED BY THE ADX, R-SQUARED, VHF, AND ER FROM APRIL 10, 2006 TO APRIL 29, 2011. The optimized test parameters are in the bottom half. Here, "TDI" is short for "trend detection indicator" for brevity's sake. The "TDI trend threshold" is the optimized trend threshold below which moving average trades were filtered out. The last three parameters (TDI crit, LAG, and MULT) are used by the second filter condition, which requires the TDI to increase MULT times from a low level to the CRIT level in LAG days. The LRS crit applies only to the R2 system and is the linear regression slope multiplied by 100.

found that a 15- to 21-day period worked best, with the threshold ranging from 0.42 to 0.30. An unusually high R-squared reading (that is, above 0.85) might indicate a possible trend reversal.

INTERPRETING THE INDICATORS

The way in which all four indicators I have described here are interpreted is similar:

- If the indicator is falling or is less than a minimum threshold, that suggests a weak trend, and you should probably use oscillators, not trend-following indicators.
- When you see the indicator rising from low levels and crossing over a certain threshold, the trend is strengthening; use trend-following systems.
- A second, higher threshold indicates a strong trend; use trend-following systems.
- An extremely high level indicates that consolidation is expected anytime. Start booking profits when oscillators indicate a downturn.
- Falling indicator readings when prices make a top, flatten out, or start declining indicate a trend reversal or the start of a consolidation period.

Trend detection indicators can be confusing because they are interpreted differently from other technical indicators, as they measure the strength but not the *direction* of the trend. This is why it's better they be used together with other technical indicators.

But which one is better at detecting trends with less lag? To compare their performance, I designed the following four trend systems using each one of the indicators I mentioned. The conclusions derived from these simulation tests can help you establish the best strategies, indicators, and parameters for trading trends.

TESTING METHOD

To test the systems, I used the TradeStation platform and the ETF United States Oil (USO). To compare the efficiency of the indicators in detecting trends, I then designed four trend-following systems using a simple 50-day moving average crossover system with each one of the four trend detection indicators. The trend detection indicators are to filter out trades generated in trendless periods according to the interpretation I described earlier. The indicators allowed trades only during established trends when the given trend detection indicator crossed over the trend threshold or when the trend detection indicator rose from low levels and crossed over a critical value (lower than the trend threshold) in a short period of time.

The second condition improved the performance of all but the R^2 indicator system. In this case, I used the linear regression slope indicator to filter out trades during low slope readings associated with sideways price movements.

The efficiency of the trend detection indicators in filtering trades was only evaluated in *initiating* the trades. All long trades were closed if the price crossed under the 50-day moving average, and the converse for short trades, irrespective of the trend detection indicator reading.

The TradeStation code for all four of these systems can be found in the sidebar "EasyLanguage Code For Trend Indicators." The indicator's period, trend level, critical level, maximum level, and so forth were determined by optimization as follows:

- Data for USO was available for the last 10 years since October 2006, so I divided the data into two five-year segments.
- I then used the first segment (in-sample data) to develop and optimize the systems and the second segment of data (out-of-sample data) to evaluate their performance. The in-sample (IS) segment span was from April 10, 2006 to April 29, 2011 and the out-of-sample (OOS) from April 29, 2011 to March 24, 2016.
- The trade size was $50,000, rounded to the nearest integer shares. Signals were executed the next day at the open, and commissions of $0.01 per share were deducted from each trade.
- The maximum number of reference bars was 50 days, so the systems started producing trades 50 days after the first bar on the chart. The IS performance and optimized parameters are shown in Figure 1.

EVALUATION OF RESULTS

The out-of-sample results (Figure 2) were rather disappointing, in sharp contrast to the illusion of excellence created by optimization of the in-sample (Figure 1) performance.

Nevertheless, all trend-filtered systems outperformed the simple moving average crossover system on a profit-factor basis, capturing the fat end of the moves and generating most of the profits from the three or four biggest trades. None of the four systems managed to beat the simple moving average system on a net-profit basis, but this was to be expected, since the number of trades produced by the trend-filtered systems was only one-fifth of the number of trades produced by the moving average system, and that's dramatically less. You shouldn't, therefore, limit your trend trading to just a couple of markets but rather search for trend opportunities on as many markets as possible if you want to avoid getting stuck in a dead-end market.

The R-squared system performed best on all metrics, managing to produce a net profit of $41,000 on only 17 trades, compared with the $44,500 profit on 84 trades of the simple moving average system. The ADX filtered out the most trades, producing only 14 trades, thus improving the percentage of profitable trades from 38% for the moving average system to 57%. The VHF was the worst performer, producing the lowest net profit and highest drawdown.

In order to evaluate the efficiency of each system in detecting strong trends, I made a list of the best trend trades produced by the moving average crossover system during the OOS period

EASYLANGUAGE CODE FOR TREND INDICATORS

ADX

```
{************************************
Strategy     : ADX
Developed by Markos Katsanos Copyright 2016
************************************}
inputs:AMT(50000),
PERIOD(14),TREND(30),ADXMAX(42),LAG(12),CRIT(22),MULT(1.8),
MAB(50);

variables: SH(0);
SH=ROUND(AMT/C,0);

// LONG
If  ADX(PERIOD) Crosses OVER  TREND AND ADX(PERIOD)<ADXMAX
AND C> AVERAGE(C,MAB)  THEN BUY("STRONG TREND") SH SHARES
NEXT BAR AT MARKET;
IF ADX(PERIOD)>MULT*LOWEST(ADX(PERIOD),LAG) AND
ADX(PERIOD) CROSSES OVER CRIT  AND C> AVERAGE(C,MAB)
THEN BUY("TREND") SH SHARES NEXT BAR AT MARKET;

//SHORT
If  ADX(PERIOD) Crosses OVER  TREND AND ADX(PERIOD)<ADXMAX
AND C< AVERAGE(C,MAB)  THEN SELLSHORT ("STRONG DNTREND")
SH SHARES NEXT BAR AT MARKET;
IF ADX(PERIOD)>MULT*LOWEST(ADX(PERIOD),LAG) AND
ADX(PERIOD) CROSSES OVER CRIT  AND C< AVERAGE(C,MAB)
THEN SELLSHORT ("DOWNTREND") SH SHARES NEXT BAR AT MAR-
KET;

//EXIT
IF C CROSSES UNDER AVERAGE(C,MAB) THEN SELL("MA") ALL
SHARES NEXT BAR AT MARKET;
IF C CROSSES OVER AVERAGE(C,MAB) THEN BUYTOCOVER("XMA")
ALL SHARES NEXT BAR AT MARKET;
If LastBarOnChart then sell("LAST")  THIS BAR AT CLOSE;
If LastBarOnChart then BUYTOCOVER ("XLAST") THIS BAR AT CLOSE;
```

R-squared

```
//Strategy: R2

inputs:AMT(50000),
R2LENGTH(18),SMOOTH(3),TREND(.42),R2MAX(.85),LAG(10),
LRCRIT(10),MAB(50);

variables: SH(0),R21(0),MAST(0),ST(0),OB(0),R2(0),R22(0),LR(0);
SH=ROUND(AMT/C,0);

//R2
If barnumber>R2LENGTH+SMOOTH+LAG then begin
R21 = COEFFICIENTR(C,CUM(1),R2Length);
R22=AVERAGE(R21,SMOOTH);
R2 = Square(R22);
LR=100*LinearRegSLOPE(C, R2Length);

If R2 CROSSES OVER TREND AND R2<R2MAX AND R2>R2[LAG] THEN
BEGIN
// LONG
If C > AVERAGE(C,MAB) AND LR>LRCRIT THEN BUY("STRONGTREND")
SH SHARES NEXT BAR AT MARKET;
//SHORT
If C <AVERAGE(C,MAB) AND LR<-LRCRIT  THEN SELLSHORT
("STRONGDOWNTREND") SH SHARES NEXT BAR AT MARKET;
END;END;
IF C CROSSES UNDER AVERAGE(C,MAB) THEN SELL("MA")
ALL SHARES  NEXT BAR AT MARKET;
IF C CROSSES OVER AVERAGE(C,MAB) THEN BUYTOCOVER("XMA")
 ALL SHARES  NEXT BAR AT MARKET;
If LastBarOnChart then sell("LAST")  THIS BAR AT CLOSE;
If LastBarOnChart then BUYTOCOVER ("XLAST")  THIS BAR AT CLOSE;
```

VHF

```
//Strategy: VHF (vertical horizontal filter)

inputs:AMT(50000),PERIOD( 36 ),TREND(.38), LAG(10),MULT(1.5),
  MAB(50),CRIT(.24),VHFMAX(.45);

variables: SH(0),VHF1(0);
SH=ROUND(AMT/C,0);
VHF1=VHF( PERIOD, H, L, C );
//LONG
If VHF1 Crosses OVER TREND AND VHF1<VHFMAX AND
VHF1>VHF1[LAG]
AND C> AVERAGE(C,MAB)  THEN BUY("STRONG TREND")
SH SHARES NEXT BAR AT MARKET;
IF VHF1>MULT*LOWEST(VHF1,LAG) AND VHF1 CROSSES OVER CRIT
AND C> AVERAGE(C,MAB)
THEN BUY("TREND") SH SHARES NEXT BAR AT MARKET;
//SHORT
If VHF1 Crosses OVER TREND AND VHF1<VHFMAX AND
VHF1>VHF1[LAG]
AND C< AVERAGE(C,MAB)  THEN SELLSHORT ("STRONG DNTREND")
SH SHARES NEXT BAR AT MARKET;
IF VHF1>MULT*LOWEST(VHF1,LAG) AND VHF1 CROSSES OVER CRIT
AND C< AVERAGE(C,MAB)
THEN SELLSHORT ("DOWNTREND") SH SHARES NEXT BAR AT MAR-
KET;
//EXIT
IF C CROSSES UNDER AVERAGE(C,MAB) THEN SELL("MA")
ALL SHARES NEXT BAR AT MARKET;
IF C CROSSES OVER AVERAGE(C,MAB) THEN BUYTOCOVER("XMA")
ALL SHARES NEXT BAR AT MARKET;
If LastBarOnChart then sell("LAST")  THIS BAR AT CLOSE;
If LastBarOnChart then BUYTOCOVER ("XLAST")  THIS BAR AT CLOSE;
```

ER

```
//Strategy: KAUFMAN'S EFFICIENCY RATIO (ER)

inputs:AMT(50000),
PERIOD(24),SMOOTH(3),TREND(.36),ERMAX(.42),ERLAG(3),
  MULT(2.5),
CRIT(.26),MAB(50);

variables: SH(0),MAST(0),ST(0),OB(0),change(0), noise(0), diff(0),
ratio(0), signal(0),ER(0);
SH=ROUND(AMT/C,0);
if currentbar > period+SMOOTH+ERLAG then begin
ratio = 0;
    diff = AbsValue(close - close[1]);
    change = close - close[period];
    signal = AbsValue(change);
                noise = summation(diff,period);
                ratio = 0;
                if noise <> 0 then ratio = signal/noise;
                ER = average(ratio,SMOOTH);
//LONG
If ER CROSSES OVER TREND AND ER<ERMAX AND
ER>ER[ERLAG]
AND C > AVERAGE(C,MAB)  THEN BUY("STRONGTREND")
SH SHARES NEXT BAR AT MARKET;
IF ER>MULT*LOWEST(ER,ERLAG) AND ER CROSSES OVER CRIT
AND C> AVERAGE(C,MAB)  THEN BUY("TREND") SH SHARES
NEXT BAR AT MARKET;
//SHORT
If ER CROSSES OVER TREND AND ER<ERMAX AND
ER>ER[ERLAG]
AND C <AVERAGE(C,MAB)   THEN SELLSHORT ("STRONGDOWN-
TREND") SH SHARES NEXT BAR AT MARKET;
IF ER>MULT*LOWEST(ER,ERLAG) AND ER CROSSES OVER CRIT
AND C< AVERAGE(C,MAB)  THEN SELLSHORT ("DOWNTREND")
SH SHARES NEXT BAR AT MARKET;
//EXIT
IF C CROSSES UNDER AVERAGE(C,MAB) THEN SELL("MA")  ALL
SHARES  NEXT BAR AT MARKET;
IF C CROSSES OVER AVERAGE(C,MAB) THEN
BUYTOCOVER("XMA")  ALL SHARES  NEXT BAR AT MARKET;
END;
If LastBarOnChart then sell("LAST")  THIS BAR AT CLOSE;
If LastBarOnChart then BUYTOCOVER ("XLAST") THIS BAR AT
CLOSE;
```

and then compared them with the trade list of the trend-filtered systems. Out of the 84 trades produced by the moving average system, only eight trades were generated during a strong trend, each trade producing more than 6% of profits in a reasonable amount of time. The rest were either marginally profitable or were whipsaw trades, most of them leading to small or moderate losses.

OUT-OF-SAMPLE PERFORMANCE (2011–2016)

System	MA	Buy & Hold	ADX	R^2	VHF	ER
Net profit	$44,592	-$38,867	$30,429	$40,968	$21,257	$26,552
Total no. of trades	84	1	14	17	18	18
Annual rate of return	13.87%	-26.36%	10.17%	12.96%	7.48%	9.06%
Percent profitable	38.10%	0.00%	57.10%	54.50%	29.40%	27.80%
Profit factor	1.86		3.99	4.18	1.93	2.25
Max losing trade	$-2,830	$-38,867	$-2,596	$-2,274	$-3,800	$-3,227
Max intraday drawdown	$-16,775	$-41,988	$-14,261	$-10,715	$-19,048	$-14,623
Avg. delay in bars (lag)	0		16	7	11	7
Missed trend trades	0		4	3	4	3
Total profit in trend trades	$77,448		$36,628	$47,425	$43,480	47,750
Avg. entry efiiciency	100.0%		56.0%	72.6%	66.0%	63.8%

FIGURE 2: OUT-OF-SAMPLE (OOS) PERFORMANCE OF THE 50-DAY SMA, BUY & HOLD, AND SYSTEMS FILTERED BY THE ADX, R-SQUARED, VHF, AND ER FROM APRIL 29, 2011 TO MARCH 24, 2016. Data was loaded from February 16, 2011, since the system required 50 trading days to calculate the indicators. The test parameters were not optimized. The last four rows display the statistics concerning the efficiency of each indicator in detecting the strong trend trades, only. The first is the lag (in addition to the 50-day MA lag) in confirming the trend and below is the number of missed trending trades (out of 8 in total), the total profit realized from these trend trades and the last row is the entry efficiency compared with the moving average entry signal.

The trend filters managed to eliminate most of the whipsaw trades, but also missed a number of the strong trends. In the last four rows of Figure 2, I calculated the statistics concerning the efficiency of each indicator in detecting the strong trend trades. First, I calculated the lag in confirming the trend, and below that is the number of missed trend trades (out of a total of eight), the total profit realized from trend trades, and in the last row, the entry efficiency compared with the profit realized by the moving average signal.

The indicators that detected trends with the least lag (seven trading days) were the R-squared and ER, with the R-squared producing the highest entry efficiency (73%), capturing most of the profits of the major moves. The overall performance of the ER, however, suffered because of four rather large whipsaw losing trades and two late entries. The ADX was the least efficient, requiring more than double the amount of time for detecting the trends.

The VHF produced the largest drawdown mainly because of two whipsaw losing trades and three late entries. In Figure 3 you can see typical filtered short trades during USO's strong downtrend at the end of 2015. The first indicator to detect the trend was the ER followed by R-squared and the VHF. The ADX was the last to detect the downtrend after a considerable delay.

A major problem with these four indicators (except maybe R-squared) is getting conflicting signals when price transitions from one direction to another in a short period of time and especially during the right leg of a V-bottom or a V-top.

Let's take a closer look at what happened during the V-bottom in June 2012 when USO made a bottom on June 28, 2012 at 29.02 and subsequently rose 25% two months later.

In the chart in Figure 4 you can see the 24-day trend detection indicators plotted in blue below the price chart. Only the R-squared detected the reversal, rising from a low of 0.01 to 0.39 on July 18, 2014 as the correlation with the 24-day linear regression line improved, especially during the last eight days. The VHF remained flat around 0.26 during the first 12 days of the reversal. Why do you think that is?

The conclusions derived from these simulation tests can help you establish the best strategies, indicators, and parameters for trading trends.



The problem is in the numerator of the VHF formula, that is, the difference between the highest and lowest price during the VHF period (24 days). In cases of V-bottoms or tops with symmetrical sides when a new point is plotted, the high at the left side is dropped and substituted by a similar value at the right side with the lowest price (at the bottom) remaining the same. The VHF therefore bottomed out at half its period (in this case, 0.5 x 24 = 12 days) after the V-bottom and started rising after that.

A similar problem exists with the ER. The numerator of the formula is the absolute value of the difference between the latest price and the price 24 days ago. That means that during a V-bottom, when a new point is plotted, the higher close at the left side is dropped and substituted by a lower close at the right side of the V-bottom, resulting in the ER declining during the first half of the period (12) days from the breakout. This is exactly what happened here when the ER declined from 0.4 at



FIGURE 3: LAG IN DETECTING THE DOWNTREND. The four trend detection indicators are compared to the moving average signals during USO's precipitous decline at the end of 2015. The efficiency ratio (ER) indicator was the first to detect the downtrend, followed by the R-squared indicator and the VHF indicator. The ADX was the last to detect the trend after a considerable delay. Trades are initiated only after the indicator crosses the trend threshold (red horizontal line). The trend threshold and indicator period were obtained by optimization of the in-sample data segment.

The indicators that detected trends with the least lag were the R-squared and the efficiency ratio.



FIGURE 4: DETECTING TREND REVERSALS. The 24-day R2, ER, VHF, and ADX are plotted below the price chart with the red horizontal line (in red) indicating the trend threshold. Notice that only the R2 crossed over the trend threshold at the time (see vertical dashed line) following the V-bottom reversal at the end of June. The VHF crossed over the threshold briefly but went down again, while the ADX actually went down indicating a weakening trend.

the breakout on June 28, 2012 to 0.05 on July 13, 2012. So in cases of V price reversals, both indicators can only start rising after a lag of at least half their period.

The ADX preformed even worse. It made a top on June 28, 2015 (at the lowest price of the V formation) and subsequently fell to 16 on August 21, 2012 at the end of the trend, missing this trade altogether.

The problem again stemmed from the numerator of the formula, which is the absolute value of the difference between +DI and –DI. As you can see in Figure 4, this difference was at its maximum during the bottom and subsequently declined, resulting in lower ADX values.

AND THE WINNER IS...

The results of this study were somewhat disappointing, as none of the trend detection indicators managed to generate more net profits than the simple moving average crossover system. They were also sometimes too late in identifying the trends, leading to losses because of sudden price reversals. Another major problem, as mentioned earlier, was bad signals during price transitions from one direction to another in a short period of time. Nevertheless, they did manage to eliminate most of the whipsaw trades that the moving average system produced during congestion periods. The R-squared indicator was the best overall performer during the out-of-sample testing, identifying trends with the least lag and eliminating most of the whipsaw trades. Kaufman's ER deserves an honorable mention because although the overall performance was not as good, it managed to detect the larger moves with less lag.

In any case, before using a trend detection indicator, you should understand how it works, understand its strength and weaknesses, and choose the best parameters (period and trend threshold) for the securities that you trade.

The strategy used for this article is not a complete trading system. It was only designed to test the efficiency of the indicators in detecting trends. Although using trend detection indicators can be a valuable starting point in analyzing trends, you should keep in mind that both the moving average and the trend detection indicators are lagging indicators. In a complete system, they should be used in conjunction with other indicators in order to reduce the lag of the entry & exit conditions. In designing such a system, you should keep in mind that the duration and strength of a trend varies across markets. Stocks tend to trend less than futures, and they revert to the mean more often. So you need to adjust your expectations and approach. The short side of equity trading

is especially challenging for standard trend-following models because as you start entering short positions on established trends, stocks tend to make vicious countermoves, which will force you out of a position.

Markos Katsanos is a STOCKS & COMMODITIES Contributing Writer and the author of Intermarket Trading Strategies. *He can be reached at markos.katsanos@gmail.com or through his website at http://mkatsanos.com.*

The code given in this article is available in the Subscriber Area at our website, www.Traders.com, in the ***Article Code*** *area.*

See our ***Traders' Tips*** *section beginning on page 50 for implementation of Katsanos' technique in various technical analysis programs. Accompanying program code can be found in the Traders' Tips area at Traders.com.*

FURTHER READING

Gopalakrishnan, Jayanthi [2000]. "Vertical Horizontal Filter," *Technical Analysis of* STOCKS & COMMODITIES, Volume 18: July.

Hartle, Thom [1991]. "The Average Directional Movement Index (ADX)," *Technical Analysis of* STOCKS & COMMODITIES, Volume 9: March.

Kaufman, Perry [2014]. "A Better Trend," *Technical Analysis of* STOCKS & COMMODITIES, Volume 32: April.

_____ [1995]. Smarter Trading: Improving Performance In Changing Markets, McGraw-Hill.

Katsanos, Markos [2009]. *Intermarket Trading Strategies*, John Wiley & Sons.

_____ [2004]."Using Money Flow To Stay With The Trend," *Technical Analysis of* STOCKS & COMMODITIES, Volume 22: June.

Star, Barbara [2007]. "Confirming Price Trend," *Technical Analysis of* STOCKS & COMMODITIES, Volume 25: December.

‡TradeStation

‡See Editorial Resource Index
†See Traders' Glossary at Traders.com for definition

